Stirling's formula

Stirling's formula provides a useful approximation of the factorial function and gamma function for very large numbers:

p! = Gamma(p+1) ~ p^p * exp(-p) * sqrt(2*pi*p)

Here, we follow the qualitative proof from Boas (Mathematical Methods in the Physical Sciences).

(%i19) `declare(p,constant);`

(%o19) done

We first start with the integrand of the definition of Gamma(p+1)

(%i20) `expr1: x^p·exp(−x)·del(x);`

expr1 $x^p \%e^{-x} \, del(x)$

(%i21) `expr2: subst('exp(log(x^p)),x^p,expr1);`

expr2 $\%e^{p \log(x) - x} \, del(x)$

Now we make the following change of variables

(%i22) `x(y) := p + y·sqrt(p);`

(%o22) $x(y) := p + y\sqrt{p}$

(%i23) `diff(x(y));`

(%o23) $\sqrt{p} \, del(y)$

(%i24) `expr3: subst([del(x)=diff(x(y)),  x=x(y) ], expr2);`

expr3 $\sqrt{p} \, \%e^{p \log(\sqrt{p}\, y + p) - \sqrt{p}\, y - p} \, del(y)$

Now we take only the exponent and expand it in a Taylor series for large p

(%i25) `dpart(expr3,2,2);`

(%o25) $\sqrt{p} \, \%e^{\boxed{p \log(\sqrt{p}\, y + p) - \sqrt{p}\, y - p}} \, del(y)$

(%i26) `exponent1: part(expr3,2,2);`

exponent1 $p \log(\sqrt{p}\, y + p) - \sqrt{p}\, y - p$

(%i27) `exponent1;`

(%o27) $p \log(\sqrt{p}\, y + p) - \sqrt{p}\, y - p$

(%i28) `onlylog1: part(exponent1,1,2);`

onlylog1 $\log(\sqrt{p}\, y + p)$

(%i29) `onlylog2: expand(taylor(onlylog1,p,inf,1));`

onlylog2 $-\left(\frac{y^2}{2\,p}\right) + \frac{y}{\sqrt{p}} + \log(p)$

(%i30) `exponent2: expand(substpart(onlylog2,exponent1,1,2));`

exponent2 $-\left(\frac{y^2}{2}\right) + p \log(p) - p$

We substitute back and integrate

(%i31) `expr4:(substpart(exponent2,expr3,2,2));`

expr4 $\sqrt{p} \, \%e^{-\left(\frac{y^2}{2}\right) + p \log(p) - p} \, del(y)$

(%i32) `integrand1: part(expr4,[1,2]);`

integrand1 $\sqrt{p} \, \%e^{-\left(\frac{y^2}{2}\right) + p \log(p) - p}$

(%i33) `expr5: ratsimp(integrate(integrand1,y,-sqrt(p),inf));`

expr5
$$\frac{\sqrt{\pi}\,\sqrt{p}\,\%e^{-p}\left(\sqrt{2}\,p^{p}\,\mathrm{erf}\left(\frac{\sqrt{p}}{\sqrt{2}}\right)+\sqrt{2}\,p^{p}\right)}{2}$$

Because

(%i34) `erf(inf);`

(%o34) 1

which is small compared to large values of p,
we make the erf(...) -> 1

(%i35) `dpart(expr5,1,4,1,3);`

(%o35)
$$\frac{\sqrt{\pi}\,\sqrt{p}\,\%e^{-p}\left(\sqrt{2}\,p^{p}\left(\boxed{\mathrm{erf}\left(\frac{\sqrt{p}}{\sqrt{2}}\right)}\right)+\sqrt{2}\,p^{p}\right)}{2}$$

(%i36) `Stirling: substpart(1,expr5,1,4,1,3);`

Stirling $\sqrt{2}\,\sqrt{\pi}\,\%e^{-p}\,p^{p+1/2}$